# COMPUTER PROGRAM FOR CALCULATING THEORETICAL CURVES AND IT'S APPLICATION TO VES DATA IN GEOTHERMAL AREAS

Keisuke USHIJIMA and Kazuo USHIJIMA

FACULTY OF ENGINEERING, KYUSHU UNIVERSITY 36
FUKUOKA 812, JAPAN

## 1. INTRODUCTION

In order to interpret Vertical Electrical Soundings (VES) data obtained in geothermal areas, a complete curve matching method may be used on the condition that the subsurface of the area is a horizontally layered structure. For this purpose, several albums of theoretical curves have been published by many authors, for instance, Mooney and Wetzel (1956), La Compagnie Generale de Geophysique (1955), Orellana and Mooney (1966), E.A.E.G. (1969) and others. However, these albums are frequently inadequate for interpreting field curves when several layers are present. They are applicable when the depth ratio of the bedrock to the thickness of the overburden is not greater than 30. This means that it is impossible to recognize the deepest structure if the thickness of the first layer is not great enough. In a geothermal field survey, it is usually encountered that the thickness of the first layer is of few meters and the depth to the bedrock is greater than 1000 meters. This is the reason why a set of theoretical curves having a maximum depth ratio of 300 or 500 must be calculated. The development of these curves requires the computations of the exact apparent resistivities for large electrode separations but these produce many difficulties in the numerical calculation.

The authors have already published two programs, PSH-A (1977a) and PSH-B (1977b), written in FORTRAN to compute theoretical curves for Wenner and Schlumberger electrode arrangements. These programs have been developed using a large scale computer (FACOM 230-45S) but they are so small in size that a geophysicist can easily make use of them on a mini-computer.

This paper describes the PSH-B program and it's practical application to the VES curves obtained in Otake (Lines L0 and L5) and Takenoyu (Lines M1 and M2) geothermal fields of Japan.

## 2. ALGORITHM OF THE PROGRAM

The apparent resistivity formulas for Wenner and Schlumberger arrangements are

$$\rho_{aW} = \rho_1 \{ 1 + \sum_{i=1}^{\infty} A(i) f_W(a,i) \} \quad (1)$$

$$\rho_{aS} = \rho_1 \{ 1 + \sum_{i=1}^{\infty} A(i) f_S(s,i) \} \quad (2)$$

where $\rho_1$ = resistivity of the first layer, a = electrode separation for Wenner array, s = electrode separation for Schlumberger array, A(i) = the i-th coefficient of series expansion of the kernel function, $f_W(a,i)$ and $f_S(s,i)$ = the distance factors expressed respectively as

$$f_W(a,i) = 2a \{ 2/(4i^2+a^2)^{1/2} - 1/(i^2+a^2)^{1/2} \} \quad (3)$$

$$f_S(s,i) = 2/(1+4i^2/s^2)^{3/2} \quad (4)$$

The summations of the eqs.(1) and (2) are computed up to a finite term. The error produced by truncating the series after N terms is estimated by Mooney et. al. (1966). According to their result, this error is no greater than $s^3/8N^2$ for Schlumberger array.

In order to minimize the errors produced during the summation process of the equation (3), it is better to rewrite as

$$f_W(a,i) = \frac{6a^3}{\sqrt{(4i^2+a^2)(i^2+a^2)}\{2\sqrt{i^2+a^2}+\sqrt{4i^2+a^2}\}} \quad (5)$$

The coefficients A(i) can be obtained recurrently (Argelo, 1967):

$$A(i) = p_{i+1} - \sum_{j=2}^{h_{n-1}} q_j A(i-j+1) \quad (6)$$

where $h_{n-1}$ = depth to the bedrock, $p_{i+1}$, $q_j$ are functions of the reflection coefficients $k_j = (\rho_{j+1} - \rho_j)/(\rho_{j+1} + \rho_j)$ and they are determined from the recursion formula for n-layer case (Flathe, 1955):

$$\left. \begin{aligned} P_n(u) &= P_{n-1}(u) + k_{n-1} u^{h_{n-1}} \{ P_{n-1}(\tfrac{1}{u}) + Q_{n-1}(\tfrac{1}{u}) \} \\ Q_n(u) &= Q_{n-1}(u) - k_{n-1} u^{h_{n-1}} Q_{n-1}(\tfrac{1}{u}) \end{aligned} \right\} \quad (7)$$

with $u = \exp(-2\lambda)$, $\lambda$ = separation constant.

Since $P_1(u) = 0$, $Q_1(u) = 1$ (8)

eqs.(7) can be expressed as

$$\left.\begin{aligned} P_n(u) &= \sum_{i=2}^{k_n-1} p_i\, u^{i-1} \\ Q_n(u) &= \sum_{i=1}^{k_n-1} q_i\, u^{i-1} \end{aligned}\right\} \qquad (9)$$

Therefore the apparent and relative resistivities can be calculated using equations (4), (5) and (6).

## 3. COMPUTER PROGRAM PSH

Argelo (1967) and Onodera (1969) have already published computer program which compute apparent resistivity curves for a general n-layer case. These programs are written in ALGOL.

The PSH program is written in FORTRAN. It is specially designed not to occupy a large memory and not to consume much time, so that field engineer can calculate theoretical curves by using a mini-computer. The PSH-A program has the following features:

(1) The computations are done in single precision arithmetic.
(2) The maximum number of polynomial coefficients is limited to 300.
(3) The summation process in equations (1) and (2) is stopped when the following condition is fullfilled:

$$\Delta\rho_{aN} = 0.00001 \cdot \rho_a \qquad (10)$$

where $\Delta\rho_{aN}$ is the difference between the latest calculated apparent resistivity (for N terms) and the preceding one (for N-1 terms), $\rho_a$ is the latest calculated apparent resistivity.

The convergence of the sums of eqs. (1) and (2) is fast for small electrode separation. For some models which converges fast, the PSH-A program is very adequate because the condition (10) is soon satisfied. However the VES curves obtained in geothermal areas do not show these characteristics, so that, the calculations are often performed to the maximum number of terms (N=300). Therefore in order to avoid the waste of time required by the condition (10), the test statement of PSH-A program should be removed. This modificated program is called PSH-B (see Appendix) and it finds its main application to the interpretation of VES curves obtained in geothermal areas. It has the follwing features:

(1) It is operated in a single precision arithmetic (7 decimal digits in length on FACOM 230-45S computer)
(2) It has not test statement given by the condition (10), so that the number of polynomial coefficients is always fixed to 300.
(3) Array variables in the DIMENSION statement (see Appendix) can be freely redimensioned to a large extent depending on the difficulty of convergence and the computer capacity.
(4) The computer output consists of 55 values of relative and apparent resistivities, calculated for 55 fixed values of electrode separations given to the program using DATA statements.
(5) All thickness of a multiple layer must be an integer value.

All executable statements in the PSH-B program are accompanied by execution numbers which are generated by FORTRAN Program Dynamic Analyzing system (Fujimura and Ushijima,1976). This execution profile shows that the total computing time is mainly dependent on the executions of eq. (6), which is proportional to $N^2$ and on the executions of eqs.(1) and (2) which are proportional to N. The former corresponds to 44850 executions which is nearly equal to $300^2/2$ while the latter corresponds to 16500 = 300 x 55 executions.

In many case studies, the total time required for computing one set of Wenner and Schlumberger curves is 14 seconds for N = 300 and 60 seconds for N = 1000 on FACOM 230-45S. Since the consumption of storage memory is directly proportional to the number of terms only, it does not mean a serious problem.

In order to keep the accuracy, the variables RWO and RSO which store the sums of the power series, are declared in double precission arithmetic. However, it is ascertained that for many cases, this declaration is not necessary.

As already pointed out, the PSH-B program can be directly executed on a mini-computer. Actually, it was tested on some mini-computers. For example, for a multiple layer model, the total computing time was 53 seconds when it was run on a mini-computer, FACOM U-200L, which has a floating point arithmetic system internally in the machine and about 9 minutes on another mini-computer, OKITAC-4300b, where the floating point arithmetic is carried out by subroutines.

## 4. APPLICATION OF PSH-B TO FIELD DATA

The PSH-B program was applied to 35 VES curves obtained in Otake area (Oita prefecture) and 29 VES curves obtained in Takenoyu area (Kumamoto prefecture) both in Kyushu Island.

### 4.1. CLASSIFICATION OF VES CURVES

The field VES curves can be classified according to their shape by the use of four types of three layer curves: type H($\rho_1 > \rho_2$

$< \rho_3$ ), type K( $\rho_1 < \rho_2 > \rho_3$ ), type A( $\rho_1 < \rho_2 < \rho_3$ ) and type Q( $\rho_1 > \rho_2 > \rho_3$ ). The resume of the classification is shown in Figure 1, where it is also found that typical VES curves observed in geothermal areas are of type H for three-layer case, types KH and QH for four-layer case and types AKH, KHA, KQH, HKH, QHA, QQH for five-layer case.




Fig. 1. Classification of VES curves

In Otake area, it is noticed that 73 % of the total curves are of types H, KH and QH on the surveying line L0 and 76 % of the total on the line L5. However, in Takenoyu area, the number of QH type curves decreases while that of HKH type does not exist. Such a classification might be very helpful and convenient for comparing geothermal areas.

## 4.2. RESULTS OF APPLICATION

It was found that 56 of the 64 VES curves show good shapes. The remaining 8 VES curves, which are four of the type KH, one of the type HA and one of the type QH, and two of the type KQH, showed a very bad shape of the apparent resistivity curve for large electrode separations. The program was applied to these 8 curves increasing the number of terms from 400 up to 1000. After these trials, it was concluded that N = 1000 gave the most satisfactory results.

## 4.3. INTERPRETATION PROCEDURES

In order to apply the PSH-B program for interpreting VES curves obtained at geothermal area, the following procedures are recommended.

(1) In order to detect faults and horizontal structures, obtain the horizontal resistivity mapping curves from the VES data.

(2) Make corrections on the VES curves if necessary, then they can be interpreted by conventional methods, for instance, using the partial curve matching method by means of the two layer standard curves and an auxilliary curve. Then rough estimations of the resistivities, number and thicknesses of the layers can be done.

(3) Input the estimated values of layer parameters to PSH-B program then theoretical curves are obtained. The field VES data are then compared with the new curves and reinterpreted.

(4) If good agreements are not reached, the procedure (3) is repeated until good fits between the theoretical and field curves are obtained.

(5) The determined subsurface structures are then drawn in the topographic section of the surveying line.

(6) The subsurface structures that indicate the possible existence of geothermal reservoirs are decided taking in considerations of various factors affecting the resistivity values.

## 5. CONCLUDING REMARKS

PSH-B program which compute the apparent and relative resistivities for a multiple layer model was presented and it was applied to 64 VES curves obtained in Otake and Takenoyu geothermal areas. It produced good theoretical curves for 56 cases, by only using 300 terms. For the remaining 8 cases, acceptable curves were obtained with the slight modification of increasing the maximum number of terms of polynomials to N = 1000.

The total computing time for the 64 curves was 64 x 14 seconds + 8 x 60 seconds = 23 minutes.

With these results, it is possible to conclude that the PSH-B program can be applied to the interpretation of the VES curves obtained in geothermal areas.

## ACKNOWLEDGMENTS

The authors wish to express their thanks to Professor S. Onodera for his helpful discussions. They wish to express thanks to Enrique Lima, Graduate Course Student of Kyushu University, for his critically reading the manuscript.

## REFERENCES CITED


Mooney, H. M., and Wetzel, W. W., 1956, The potentials about a point electrode and apparent resistivity curves for a two-, three- and four-layered earth: One volume and one atlas. University of Minnesota Press, Mineapolis, Minnesota.

La Compagnie Generale de Geophysique, 1955, Abaques de sondage electrique: Geophysical Prospecting, v.3, Supplement,no.3.

Orellana, E., and Mooney, H. M., 1966, Master tables and curves for vertical electrical sounding over layered structures: Interciencia, Madrid.

Rijkswaterstaat, The Netherlands, 1969, Standard graphs for resistivity prospecting, European Association of Exploration Geophysicists.

Keisuke USHIJIMA and Kazuo USHIJIMA, 1977a, Calculation of resistivity sounding curves for a horizontally layered earth: BUTSURI-TANKO, v.30, no.1, pp.1-11.

Keisuke USHIJIMA and Kazuo USHIJIMA, 1977b, Application of the computer program PSH to the interpretation of resistivity soundings in geothermal areas: BUTSURI-TANKO, v.30, no.4, pp.10-19.

Argelo, S. M., 1967, Two computer programs for the calculation of standard graphs for resistivity prospecting, Geophysical Prospecting, v.15, no.1, pp.71-91.

Seibe ONODERA, 1969, Interpretation of VES curves by digital computer, Journal of the Mining and Metallurgical Institute of Japan, v.85, no.976, pp.7-10.

Mooney, H. M., Orellana, E., Pickett, H., and Tornheim, L., 1966, A resistivity computation method for layered earth models: Geophysics, v.31, no.1, pp.192-203.

Flathe, H., 1955, A practical method of calculating geoelectrical model graphs for horizontally stratified media: Geophysical Prospecting, v.3, pp.38-44.

Naomi FUJIMURA and Kazuo USHIJIMA, 1976, On transferability of Fortran dynamic analyzing system, JYOHO-SHORI, v.17, no. 11, pp.1048-1055.

Keller, G. V., and Frischknecht, F. C., 1970, Electrical methods in geophysical prospecting: Pergamon press, pp.135-136.

Takashi NOGUCHI and Seibe ONODERA, 1969, Resistivity exploration for altered zone at Otake geothermal area - Japan: Bulletin Volcanologique, Tome 33-1, pp. 205-228.


## APPENDIX

```
      FACOM 230-45S  OS2   FORDAP   -760805-   V-01  L-08   DATE 77-05-16   TIME
 * SOURCE STATEMENT *          EXECUTION TIME =   15643 MSEC        EXECUTIONS

C**** WENNER AND SCHLUMBERGER RESISTIVITY SOUNDING CURVES FOR
C****               HORIZONTALLY LAYERED EARTH
C
C     PROGRAM PSH-B CODED BY KEISUKE USHIJIMA AND KAZUO USHIJIMA
C     FACULTY OF ENGINEERING, KYUSHU UNIVERSITY, 6-10-1, HAKOZAKI
C     FUKUOKA 812, JAPAN
C     CRITICISM INVITED  TEL 092-641-1101 EXT.3218
C     VERSION OF MAY 10, 1977
C
      INTEGER STAN
      DIMENSION RHO(10),IM(9),STAN(10)
C
C**** CHANGE FOLLOWING TWO STATEMENTS IF ADVANCED CALCULATION NEEDED.
C
      DIMENSION P( 301),Q( 301),COEF( 300)
      N=300                                                            1
C
      WRITE(6,10)                                                      1
   10 FORMAT(1H1,/,10X,'RESISTIVITY SOUNDING CURVES BY PROGRAM PSH',/)
C
C**** INPUT OF LAYER PARAMETERS
C
   20 READ(5,30)LAYER,STAN                                             2
   30 FORMAT(I10,10A2)
      IF(LAYER .EQ. 0) STOP                                            2
      READ(5,40)(RHO(I),I=1,LAYER)                                     1
   40 FORMAT(8F10.0)
      LM1=LAYER-1                                                      1
      READ(5,50)(IM(I),I=1,LM1)                                        1
   50 FORMAT(8I10)
C
      CALL PSH(LAYER,STAN,RHO,IM,N,P,Q,COEF,N+1)                       1
C
      GO TO 20                                                         1
      END

 * SOURCE STATEMENT *          EXECUTION TIME =   15633 MSEC        EXECUTIONS

      SUBROUTINE PSH(LAYER,STAN,RHO,IM,N,P,Q,COEF,NDIM)                1
      INTEGER TERM,STAN
      INTEGER*4 T1,T2,T3
      DOUBLE PRECISION RWO,RSO
      DIMENSION A(55),AO(55),RHO(10),RK(9),IM(9),
     *          P(NDIM),Q(NDIM),COEF(N),STAN(10)
      DATA NA/55/
C
C**** INPUT OF WENNER ELECTRODE SEPARATIONS
C
      DATA A/0.1,   0.2,   0.3,   0.4,   0.5,   0.6,   0.7,   0.8,
     1      0.9,   1.0,   1.2,   1.4,   1.6,   1.8,   2.0,   2.2,
     2      2.4,   2.6,   2.8,   3.0,   3.4,   4.0,   4.4,   5.0,
     3      5.6,   6.0,   7.0,   8.0,   9.0,  10.0,  12.0,  14.0,
     4     16.0,  18.0,  20.0,  25.0,  30.0,  32.0,  40.0,  48.0,
     5     56.0,  64.0,  80.0,  96.0, 100.0, 120.0, 160.0, 200.0,
     6    240.0, 280.0, 320.0, 400.0, 480.0, 560.0, 640.0/
C
C**** INPUT OF SCHLUMBERGER ELECTRODE SEPARATIONS
C
      DATA AO/0.3,  0.4,   0.5,   0.6,   0.7,   0.8,   0.9,   1.0,
     1      1.2,   1.4,   1.6,   1.8,   2.0,   2.2,   2.5,   2.8,
     2      3.0,   3.4,   4.0,   4.4,   5.0,   6.0,   7.0,   8.0,
     3      9.0,  10.0,  12.0,  14.0,  16.0,  18.0,  20.0,  22.0,
     4     25.0,  28.0,  30.0,  34.0,  40.0,  44.0,  50.0,  60.0,
     5     70.0,  80.0,  90.0, 100.0, 120.0, 140.0, 160.0, 180.0,
     6    200.0, 220.0, 250.0, 280.0, 320.0, 380.0, 460.0/
C
      CALL CLOCKM(T1)                                                  1
C
C**** CALCULATION OF REFLECTION COEFFICIENTS
C
      LM1=LAYER-1                                                      1
      DO 60 I=1,LM1                                                    1
         RK(I)=(RHO(I+1)-RHO(I))/(RHO(I+1)+RHO(I))                     2
   60 CONTINUE                                                         2
C
C**** PRINT OF LAYER PARAMETERS
C
      WRITE(6,65)STAN                                                  1
   65 FORMAT(1H0,20X,10A2)
      WRITE(6,70)                                                      1
   70 FORMAT(1H0,8X,'LAYER',7X,'RHO',14X,'K',13X,'H',/)
      WRITE(6,80)(I,RHO(I),RK(I),IM(I),I=1,LM1)                        1
   80 FORMAT(1H ,7X,I5,F15.8,F15.8,I10)
      WRITE(6,90)LAYER,RHO(LAYER)                                      1
   90 FORMAT(1H ,7X,I5,F15.8,///)

C
C**** DETERMINATION OF P(J) AND Q(J)
C
      P(1)=0                                                           1
      Q(1)=1.0                                                         1
      DO 100 I=1,N                                                     1
         P(I+1)=0                                                    300
         Q(I+1)=0                                                    300
  100 CONTINUE                                                       300
      ID=1                                                             1
      DO 120 I=1,LM1                                                   1
         IDMIN=ID                                                      2
         ID=IM(I)+1                                                    2
         JS=ID-IDMIN+1                                                 2
         DO 110 J=JS,ID                                                2
            JE=ID-J+1                                                  3
            P(J)=P(J)+RK(I)*(P(JE)+Q(JE))                              3
            Q(J)=Q(J)-RK(I)*Q(JE)                                      3
  110    CONTINUE                                                      3
  120 CONTINUE                                                         2
C
C**** PRINT OF TITLE
C
      WRITE(6,130)                                                     1
  130 FORMAT(1H0,15X,'WENNER CURVE',30X,'SCHLUMBERGER CURVE',//,
     1       8X,'A',10X,'R(A)',10X,'AR(A)',13X,'AO',8X,
     2       'R(AO)',10X,'AR(AO)',10X,'TERM',//)
C
C**** DETERMINATION OF COEFFICIENTS
C
      COEF(1)=P(2)                                                     1
      DO 150 TERM=2,N                                                  1
         C=P(TERM+1)                                                 299
         DO 140 J=2,TERM                                             299
            IJ=TERM+1-J                                            44850
            C=C-Q(J)*COEF(IJ)                                      44850
  140    CONTINUE                                                  44850
         COEF(TERM)=C                                                299
  150 CONTINUE                                                       299
C
C**** CALCULATION OF RESISTIVITY SOUNDING CURVES
C
      DO 190 I=1,NA                                                    1
         RWO=1.0                                                      55
         RSO=1.0                                                      55
         AA=A(I)*A(I)                                                 55
         SS=AO(I)*AO(I)                                               55
         DO 160 TERM=1,N                                              55
            FJ=TERM                                                16500
            FFJ=FJ*FJ                                              16500
            FS=4.0*FFJ                                             16500
C     WENNER CURVE
            AC=6.0*AA*A(I)                                         16500
            SQA=SQRT(FS+AA)                                        16500
            SQB=SQRT(FFJ+AA)                                       16500
            WO=COEF(TERM)*AC/(SQA*SQB*(2.0*SQB+SQA))               16500
            RWO=RWO+WO                                             16500
C     SCHLUMBERGER CURVE
            SO=2.0*COEF(TERM)/(1.0+FS/SS)/SQRT(1.0+FS/SS)          16500
            RSO=RSO+SO                                             16500
  160    CONTINUE                                                  16500
         TERM=N                                                       55
         RW=RWO*RHO(1)                                                55
         RS=RSO*RHO(1)                                                55
C     PRINT OF WENNER AND SCHLUMBERGER CURVES
         WRITE(6,180)A(I),RWO,RW,AO(I),RSO,RS,TERM                    55
  180    FORMAT(1H ,F10.1,2F15.6,F15.1,2F15.6,I10)
  190 CONTINUE                                                        55
C
C**** TIME COUNT FOR ONE SET OF CURVE
C
      CALL CLOCKM(T2)                                                  1
      T3=T2-T1                                                         1
      WRITE(6,200)T3                                                   1
  200 FORMAT(1H0,//,5X,'TIME COUNT FOR NS CURVES=',I10,2X,'MS',//)
C
C**** PRINT OF COEFFICIENTS OF POWER SERIES
C
      WRITE(6,210)(JJD,COEF(JJD),JJD=1,TERM)                           1
  210 FORMAT(5(I5,F15.6))
      RETURN                                                           1
      END
```